Flux and Gauss' Divergence Theorem

Ex1.

Let $V(x,y,z) = \begin{bmatrix} \cos x \\ \cos x + \cos z \\ 0 \end{bmatrix}$, and let $\mathcal{F}$ be a surface given

by the parametric representation $r(u,v) = \begin{bmatrix} u \\ 0 \\ v \end{bmatrix}$, $u \in [0,\pi]$, $v \in [0,2]$.

a) Determine the normal vector $N_{\mathcal{F}}$ of the surface, and compute the flux of V through the surface.

$$r'_u = \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \quad r'_v = \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \quad \Rightarrow \quad N_{\mathcal{F}} = r'_u \times r'_v = \begin{bmatrix} 0 \\ -1 \\ 0 \end{bmatrix}.$$

Now just use either definition 28.1 or the equivalent of proposition 28.2.

$$\int_{\mathcal{F}} V \cdot n_{\mathcal{F}} \, d\mu = \int_0^2 \int_0^\pi V(r(u,v)) \cdot N_{\mathcal{F}} \, du \, dv$$

$$= \int_0^2 \int_0^\pi \begin{bmatrix} \cos u \\ \cos u + \cos v \\ 0 \end{bmatrix} \cdot \begin{bmatrix} 0 \\ -1 \\ 0 \end{bmatrix} du \, dv$$

$$= \int_0^2 \int_0^\pi -\cos u - \cos v \; du \, dv$$

$$= \int_0^2 \Big[-\sin u - \cos(v)\, u \Big]_0^\pi \, dv$$

$$= \int_0^2 -\cos(v) \cdot \pi \; dv$$

$$= -\pi \Big[\sin v \Big]_0^2 = -\pi \cdot \sin 2.$$

b) What is the meaning of the sign? Does the flux change if we alter the parametrization?

The normal vector and the vector field form an obtuse angle such that the dot product is negative. Another normal $\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$ yields the flux $\pi \cdot \sin 2$.

Alternative parametrization: $s(u,v) = \begin{bmatrix} u \\ 0 \\ 2-v \end{bmatrix}$, $u \in [0,\pi]$, $v \in [0,2]$.

Here $s(u,v)$ serves as a parametrization for which the flux of V through $\mathcal{F}$ is $\pi \cdot \sin 2$.

Now we let $V(x,y,z) = \begin{bmatrix} yz \\ -xz \\ x^2+y^2 \end{bmatrix}$ and $r(u,v) = \begin{bmatrix} u \sin v \\ -u \cos v \\ uv \end{bmatrix}$, $u,v \in [0,1]$.

c) Get the normal of r and compute the flux.

$$r_u' = \begin{bmatrix} \sin v \\ -\cos v \\ v \end{bmatrix}, \quad r_v' = \begin{bmatrix} u \cos v \\ u \sin v \\ u \end{bmatrix}, \quad r_u' \times r_v' = \begin{bmatrix} -u(\cos v + v \sin v) \\ u(v \cos v - \sin v) \\ u \end{bmatrix}$$

$$N_{\mathcal{F}}(u,v) = \begin{bmatrix} -u(\cos v + v \sin v) \\ u(v \cos v - \sin v) \\ u \end{bmatrix} \quad \text{and} \quad V(r(u)) = \begin{bmatrix} -u^2 v \cos v \\ -u^2 v \sin v \\ u^2 \end{bmatrix}.$$

$$\begin{aligned} V(r(u)) \cdot N_{\mathcal{F}}(u,v) &= u^3 v (\cos^2 v + v \cos v \cdot \sin v) \\ &\quad - u^3 v (v \cos v \cdot \sin v - \sin^2 v) + u^3 \\ &= u^3 v + u^3 = u^3 (v+1). \end{aligned}$$

$$\int_0^1 \int_0^1 u^3 (v+1) \, du \, dv = \int_0^1 (v+1) \cdot \frac{1}{4} \, dv = \frac{1}{8} + \frac{1}{4} = \frac{3}{8}.$$

Ex 2. Let

$$V(x,y,z) = \begin{bmatrix} x/2 \\ y/2 \\ 2z \end{bmatrix}.$$

a) Determine the flow curve $r(t)$ of V for which $r(0) = (1,1,1)$.

```
> V:=(x,y,z)-> <x/2,y/2,2*z>:
  lign1:=diff(x(t),t)=x(t)/2;
  lign2:=diff(y(t),t)=y(t)/2;
  lign3:=diff(z(t),t)=2*z(t);
```

$$lign1 := \frac{d}{dt}\, x(t) = \frac{x(t)}{2}$$

$$lign2 := \frac{d}{dt}\, y(t) = \frac{y(t)}{2}$$

$$lign3 := \frac{d}{dt}\, z(t) = 2\, z(t)$$

```
> sol:=dsolve({lign1,lign2,lign3,x(0)=1,y(0)=1,z(0)=1},{x(t),y
  (t),z(t)}):
  r:=<rhs(sol[1]),rhs(sol[2]),rhs(sol[3])>;
```

$$r := \begin{bmatrix} e^{\frac{t}{2}} \\ e^{\frac{t}{2}} \\ e^{2t} \end{bmatrix}$$

The surface S_0 is the part of the unit sphere that is above or intersects $z = \frac{1}{2}$.

b) Parametrize S_0.

We just go down $\frac{\pi}{3}$ radians.

```
> S:=<sin(u)*cos(v),sin(u)*sin(v),cos(u)>:
> sol:=dsolve({lign1,lign2,lign3,x(0)=S[1],y(0)=S[2],z(0)=S[3]},
  {x(t),y(t),z(t)}):
  r:=unapply(<rhs(sol[1]),rhs(sol[2]),rhs(sol[3])>,[u,v,t]):
  r(u,v,t);
```

$$\begin{bmatrix} \sin(u)\cos(v)\, e^{\frac{t}{2}} \\ \sin(u)\sin(v)\, e^{\frac{t}{2}} \\ \cos(u)\, e^{2t} \end{bmatrix}$$

c) Compute the flux.

```
> unassign('S0'):
  VectorCalculus[Jacobian](r(u,v,0),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  n:=simplify(%):
  integrand:=prik(V(vop(r(u,v,0))),n):
  Flux(V,S0)=int(integrand,[u=0..Pi/3,v=0..2*Pi]);
```

$$Flux(V, S0) = \frac{11\pi}{8}$$

d) Plot S_0 and the deformation over some time t.

```
> vfield:=fieldplot3d(V(x,y,z),x=-1.2..1.2,y=-1.2..1.2,z=0.5..1.2,grid=[4,4,4],color=red,arrows=SLIM):
  S0:=plot3d(r(u,v,0),u=0..Pi/3,v=-Pi..Pi,transparency=0,style=patchnogrid):
  St:=plot3d(r(u,v,0.3),u=0..Pi/3,v=-Pi..Pi,transparency=0.4,style=wireframe):
  display(vfield,axes=normal,orientation=[25,60],S0,St,scaling=constrained,view=0..2,tickmarks=[3,3,3]);
```



e) Justify that S_0 and S_t have no points in common.

```
> prik(r(u,v,t),r(u,v,t)):
  simplify(%);
```

$$-e^{t}\cos(u)^2 + \cos(u)^2 e^{4t} + e^{t}$$

$$= \cos(u)^2 e^{4t} + \sin(u)^2 e^{t}$$

We have the minimum of 1 for $t=0$, while $e^t>1$ for $t>0$.

The radius of S_0 is 1, so to speak. Thus every point of S_t for $t>0$ flows some distance from S_0.

f) Parametrize the spacial region Ω_t that S_t flows through.

```
> R:=unapply(r(u,v,w*t),[u,v,w]):
R(u,v,w);
```

$$\begin{bmatrix} \sin(u)\cos(v)\,e^{\frac{w t}{2}} \\ \sin(u)\sin(v)\,e^{\frac{w t}{2}} \\ \cos(u)\,e^{2 w t} \end{bmatrix}$$

g) Compute the volume $Vol(t)$ of Ω_t.

```
> VectorCalculus[Jacobian](R(u,v,w),[u,v,w]):
Determinant(%):
Jacobi:=simplify(%):
vol:=int(Jacobi,[u=0..Pi/3,v=0..2*Pi,w=0..1]);
```

$$vol := \frac{11\,\pi\left(-1+e^{3t}\right)}{24}$$

h) Determine $Vol'(t)$ and $Vol'(0)$. Compare with the flux.

```
> diff(vol,t);
subs(t=0,%):
simplify(%);
```

$$\frac{11\,\pi\,e^{3t}}{8}$$

$$\frac{11\,\pi}{8}$$

This corresponds to the flux, and this follows by proposition 28.6.

Ex 3. Let
$$V(x,y,z) = \begin{bmatrix} xyz \\ x+y+z \\ z/2 \end{bmatrix} \quad \text{and} \quad \alpha : z + x = 2.$$

a) Parametrize the part of α bounded by $(1,1,0)$, $(-1,-1,0)$, $(-1,-1,0)$ and $(1,-1,0)$.
Do this so that the normal has a positive third coordinate.

$$r(u,v) = \begin{bmatrix} u \\ v \\ 2-u \end{bmatrix}, \quad u \in [-1,1], \quad v \in [-1,1]. \quad (\text{Note from } \alpha \quad z = 2 - x).$$

Also $\underline{n} = \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}$.

```
> V:=(x,y,z)-> <x*y*z,x+y+z,z/2>:
  r:=(u,v)-> <u,v,2-u>:
  S:=plot3d(r(u,v),u=-1..1,v=-1..1,axes=normal,scaling=constrained,orientation=[-60,60]):
  vfield:=fieldplot3d(V(x,y,z),x=-1..1,y=-1..1,z=0..3,grid=[4,4,4],arrows=SLIM,color=red):
  display(S,vfield);
```



b) Compute the flux.

```
> unassign('S'):
  VectorCalculus[Jacobian](r(u,v),[u,v]):
  n:=kryds(Column(%,1),Column(%,2)):
  prik(V(vop(r(u,v))),n):
  integrand:=simplify(%):
  Flux(V,S)=int(integrand,[u=-1..1,v=-1..1]);
```

$$Flux(V,S) = 4$$

A surface $\mathcal{F}$ is given in two parts.

- $\mathcal{F}_1$ is the disc of α bounded by $x^2+y^2 \leq 1$.
- $\mathcal{F}_2$ is the cylinder surface bounded below by the unit circle and bounded above by α.

c) Parametrize $\mathcal{F}_1$ and $\mathcal{F}_2$ so that their normals are outward pointing.

```
> r:=(u,v)-> <u*cos(v),u*sin(v),2-u*cos(v)>:
  s:=(u,v)-> <cos(u),sin(u),v*(2-cos(u))>:
  r(u,v),
  s(u,v);
```

$$\begin{bmatrix} u\cos(v) \\ u\sin(v) \\ 2-u\cos(v) \end{bmatrix}, \begin{bmatrix} \cos(u) \\ \sin(u) \\ v(2-\cos(u)) \end{bmatrix}$$

d) Compute the flux through F.

```
> VectorCalculus[Jacobian](r(u,v),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  nr:=simplify(%);
  prik(V(vop(r(u,v))),nr):
  integrand:=simplify(%):
  Flux(V,Fr)=int(integrand,[u=0..1,v=0..2*Pi]);
```

$$nr := \begin{bmatrix} u \\ 0 \\ u \end{bmatrix}$$

$$Flux(V, Fr) = \pi$$

```
> VectorCalculus[Jacobian](s(u,v),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  ns:=simplify(%);
  prik(V(vop(s(u,v))),ns):
  integrand:=simplify(%):
  Flux(V,Fs)=int(integrand,[u=0..2*Pi,v=0..1]);
```

$$ns := \begin{bmatrix} -\cos(u)\,(-2+\cos(u)) \\ \sin(u)\,(2-\cos(u)) \\ 0 \end{bmatrix}$$

$$Flux(V, Fs) = 2\pi$$

```
> (4.4.1)+(4.4.2)
```

$$Flux(V, Fr) + Flux(V, Fs) = 3\pi$$

```
> F1:=plot3d(r(u,v),u=0..1,v=0..2*Pi,axes=normal,scaling=constrained,orientation=[-60,60]):
  F2:=plot3d(s(u,v),u=-Pi..Pi,v=0..1,axes=normal,scaling=constrained,orientation=[-60,60]):
  display(<F1|F2>);
```



```
> display(F1,F2,orientation=[25,60]);
```



Now F is rotated by ω around the z axis.

e) Compute the ω that yields max/min flux. State the values.

```
> rot:=<<cos(w),sin(w),0>|<-sin(w),cos(w),0>|<0,0,1>>:
  r1:=rot.(r(u,v)):
  n1:=rot.nr:
  r2:=rot.(s(u,v)):
  n2:=rot.ns:
> integrand1:=prik(V(vop(r1)),n1):
  int(integrand1,[u=0..1,v=0..2*Pi]):
  Flux1:=simplify(%);
```

$$Flux1 := \pi\,(1 + 2\sin(w))$$

```
> integrand2:=prik(V(vop(r2)),n2):
  int(integrand2,[u=0..2*Pi,v=0..1]):
  Flux2:=simplify(%);
```

$$Flux2 := \pi\left(2 - \frac{5\sin(w)}{2}\right)$$

```
> Flux:=unapply(Flux1+Flux2,w):
  Flux(w):
  normal(%);
```

$$-\frac{\sin(w)\,\pi}{2} + 3\pi$$

```
> Max=Flux(-Pi/2);
  Min=Flux(Pi/2);
```

$$Max = \frac{7\pi}{2}$$

$$Min = \frac{5\pi}{2}$$

Ex 4. Let Ω denote the unit cube.

a) Compute the flux of V through $\partial\Omega$, where

$$V(x,y,z) = \begin{bmatrix} 2x - \sqrt{1+z^2} \\ x^2 y \\ -xz^2 \end{bmatrix}.$$

The divergence theorem 28.15 let's us do this by means of divergence.

$$\mathrm{Div}(V)(x,y,z) = 2 + x^2 - 2xz$$

$$\mathrm{Flux}(V, \partial\Omega) = \int_\Omega \mathrm{Div}(V)\, d\mu = \int_0^1\int_0^1\int_0^1 2 + x^2 - 2xz \; dx\, dy\, dz$$

$$= \int_0^1 2 + \tfrac{1}{3} - z \, dz = 2 + \tfrac{1}{3} - \tfrac{1}{2} = \frac{11}{6}.$$

b) Compute the flux for $W(x,y,z) = \begin{bmatrix} 2x - \sqrt[3]{y^2+z^2} \\ xz - \cos y \\ \sin(xy) + 2z \end{bmatrix}.$

$$\mathrm{Div}(W)(x,y,z) = 2 + \sin y + 2 = 4 + \sin y$$

$$\mathrm{Flux}(W, \partial\Omega) = \int_0^1\int_0^1\int_0^1 4 + \sin y \; dx\, dy\, dz$$

$$= \int_0^1 4 + \sin y \; dy = \left[4y - \cos(y)\right]_0^1$$

$$= 4 - \cos(1) - (0 - \cos(0)) = 5 - \cos(1).$$

c) Given that $\int_0^1\int_0^1\int_0^1 x + y + z \; dx\, dy\, dz = \frac{3}{2}$ determine a vector field such that $\mathrm{Flux}(U, \partial\Omega) = \frac{3}{2}$.

Just set $U(x,y,z) = \frac{1}{2}\begin{bmatrix} x^2 \\ y^2 \\ z^2 \end{bmatrix}$, and add a constant if you want.

Ex5.

Let $V(x,y,z) = \begin{bmatrix} -8x \\ 8 \\ 4z^3 \end{bmatrix}$ and $\Omega = \{(x,y,z) \in \mathbb{R}^3 \mid x^2+y^2+z^2 \le a^2,\ z \ge 0\},\ a>0.$

a) Compute $\int_\Omega \mathrm{Div}(V)\, d\mu$.

```
> V:=(x,y,z)-> <-8*x,8,4*z^3>:
  r:= (u,v,w)-> <u*sin(v)*cos(w), u*sin(v)*sin(w), u*cos(v)>:
  VectorCalculus[Jacobian](r(u,v,w),[u,v,w]):
  Determinant(%):
  Jacobi:=simplify(%);
```

$$Jacobi := u^2 \sin(v)$$

```
> divV:=unapply(div(V)(x,y,z),[x,y,z]):
  int(divV(vop(r(u,v,w)))*Jacobi,[u=0..a,v=0..Pi/2,w=0..2*Pi]);
```

$$\frac{8 a^3 (3 a^2 - 10)\pi}{15}$$

b) Compute $\int_{\partial\Omega} V \cdot \underline{n}_{\partial\Omega}\, d\mu$.

```
> r(a,v,w);
  VectorCalculus[Jacobian](r(a,v,w),[v,w]):
  kryds(Column(%,1),Column(%,2)):
  n1:=simplify(%);
  Flux1:=int(prik(V(vop(r(a,v,w))),n1),[v=0..Pi/2,w=0..2*Pi]);
```

$$\begin{bmatrix} a\sin(v)\cos(w) \\ a\sin(v)\sin(w) \\ a\cos(v) \end{bmatrix}$$

$$n1 := \begin{bmatrix} a^2\sin(v)^2\cos(w) \\ a^2\sin(v)^2\sin(w) \\ a^2\cos(v)\sin(v) \end{bmatrix}$$

$$Flux1 := \frac{8}{5}a^5\pi - \frac{16}{3}a^3\pi$$

```
> r(u,Pi/2,w);
  VectorCalculus[Jacobian](r(u,Pi/2,w),[u,w]):
  kryds(Column(%,1),Column(%,2)):
  n2:=simplify(%);
  Flux2:=int(prik(V(vop(r(u,Pi/2,w))),n2),[u=0..a,w=0..2*Pi]);
```

$$\begin{bmatrix} \cos(w)\,u \\ \sin(w)\,u \\ 0 \end{bmatrix}$$

$$n2 := \begin{bmatrix} 0 \\ 0 \\ u \end{bmatrix}$$

$$Flux2 := 0$$

```
> Flux:=Flux1+Flux2
```

$$Flux := \frac{8}{5}a^5\pi - \frac{16}{3}a^3\pi$$

c) For which a is $\mathrm{Flux}(V, \partial\Omega)$ with the given unit normal vector field $\underline{n}_{\partial\Omega}$ positive.

```
> solve(Flux=0,a);
  plot(Flux,a=0..2,view=-20..30);
```

$$0, 0, 0, \frac{\sqrt{30}}{3}, -\frac{\sqrt{30}}{3}$$



d) Discuss the relation between divergence and flux integrals and $\left[F(x)\right]_a^b = \int_a^b F'(x)\,dx$.

We can think of the divergence as the derivative of the vector field, and we deal with the integration for boundary values similar to $F(b) - F(a)$.

Exb. The Coulomb vector field is given by

$$V(x,y,z) = \frac{1}{(x^2+y^2+z^2)^{3/2}} \begin{bmatrix} x \\ y \\ z \end{bmatrix}.$$

Let Ω be the solid cylinder given by

$$r(u,v,w) = \begin{bmatrix} u \cos w \\ u \sin w \\ v \end{bmatrix}, \quad u \in [0,a],\ v \in [-h,h],\ w \in [-\pi,\pi].$$

a) Draw and parametrize Ω in parts.

For $a = h = 5$:

```
> V:=(x,y,z)-> <x,y,z>/(x^2+y^2+z^2)^(3/2):
> Stop:=(u,v)-> <u*cos(v),u*sin(v),h>:
  Sbot:=(u,v)-> <u*cos(v),u*sin(v),-h>:
  Scyl:=(u,v)-> <a*cos(v),a*sin(v),u>:
> plot3d([subs(a=5,Scyl(u,v)),subs(h=5,Stop(u,v)),subs(h=5,Sbot(u,v))],u=-5..5,v=0..2*Pi,axes=boxed,scaling=constrained,
  orientation=[25,60],transparency=0.5);
```



b) Determine the flux through all parts of the boundary. How does the size influence the strength of the flux?

```
> assume(a>0,h>0): #neyðugt so at maple kann rokna integralini
  interface(showassumed=0):
> VectorCalculus[Jacobian](Stop(u,v),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  ntop:=simplify(%):
  Fluxtop:=int(prik(V(vop(Stop(u,v))),ntop),u=0..a,v=0..2*Pi);
```

$$Fluxtop := \frac{2\left(\sqrt{a^2+h^2}-h\right)\pi}{\sqrt{a^2+h^2}}$$

```
> VectorCalculus[Jacobian](Sbot(u,v),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  nbot:=-simplify(%): #vendi normalvektorin út úr flatanum
  Fluxbot:=int(prik(V(vop(Sbot(u,v))),nbot),[u=0..a,v=0..2*Pi]);
```

$$Fluxbot := \frac{2\left(\sqrt{a^2+h^2}-h\right)\pi}{\sqrt{a^2+h^2}}$$

```
> VectorCalculus[Jacobian](Scyl(u,v),[u,v]):
  kryds(Column(%,1),Column(%,2)):
  ncyl:=-simplify(%): #vendi normalvektorin út úr flatanum
  Fluxcyl:=int(prik(V(vop(Scyl(u,v))),ncyl),[u=-h..h,v=0..2*Pi]);
```

$$Fluxcyl := \frac{4\,h\,\pi}{\sqrt{a^2+h^2}}$$

```
> Flux=simplify(Fluxtop+Fluxbot+Fluxcyl);
```

$$Flux = 4\,\pi$$

The flux is constant, so it is independent of size.

c) Compute the flux by means of the divergence.

```
> div(V)(x,y,z);
  simplify(%);
```

$$-\frac{3\,x^2}{\left(x^2+y^2+z^2\right)^{5/2}}+\frac{3}{\left(x^2+y^2+z^2\right)^{3/2}}-\frac{3\,y^2}{\left(x^2+y^2+z^2\right)^{5/2}}-\frac{3\,z^2}{\left(x^2+y^2+z^2\right)^{5/2}}$$

$$0$$

So this gives us a flux of 0.

d) What went wrong?

There's a singularity at (0,0,0), and Gauss' Divergence Theorem demands a smooth vector field.