Vector Fields

Ex 1. A linear vector field $\mathcal{U}$ is given by

$$\mathcal{U}(x,y) = \begin{bmatrix} \frac{1}{8}x + \frac{3}{8}y \\ \frac{3}{8}x + \frac{1}{8}y \end{bmatrix}.$$

a) Determine $\underline{\underline{A}}$ s.t. $\mathcal{U} = \underline{\underline{A}}\begin{bmatrix} x \\ y \end{bmatrix}$, and use Maple to get the eigenvalues and eigenvectors.

We set $\underline{\underline{A}} = \begin{bmatrix} \frac{1}{8} & \frac{3}{8} \\ \frac{3}{8} & \frac{1}{8} \end{bmatrix}$ and with Maple we have

```
> Eigenvectors(A,output=list);
```

$$\left[\left[\frac{1}{2}, 1, \left\{\begin{bmatrix} 1 \\ 1 \end{bmatrix}\right\}\right], \left[-\frac{1}{4}, 1, \left\{\begin{bmatrix} -1 \\ 1 \end{bmatrix}\right\}\right]\right]$$

b) Let $r_1(u)$ be the flow curve through $(0,-1)$ when $u=0$, and let $r_2(u)$ be the flow curve through $(0, \frac{1}{2})$ when $u=0$.

State a parametric representation for the curves.

From definition 26.11 we have

$$V(r(u)) = r'(u) \quad \forall u,$$

where existence and uniqueness is given by proposition 26.12.

We solve the differential equations as in ch. 17, see method 17.4.

$$\left[\begin{array}{cc|c} 1 & -1 & 0 \\ 1 & 1 & -1 \end{array}\right] \Rightarrow \begin{bmatrix} -\frac{1}{2} \\ -\frac{1}{2} \end{bmatrix}, \qquad \left[\begin{array}{cc|c} 1 & -1 & 0 \\ 1 & 1 & \frac{1}{2} \end{array}\right] \Rightarrow \begin{bmatrix} \frac{1}{4} \\ \frac{1}{4} \end{bmatrix}$$

$$r_1(u) = -\frac{1}{2} \cdot e^{\frac{1}{2}u} \cdot \begin{bmatrix} 1 \\ 1 \end{bmatrix} - \frac{1}{2} \cdot e^{-\frac{1}{4}u} \cdot \begin{bmatrix} -1 \\ 1 \end{bmatrix}$$

$$= \begin{bmatrix} -\frac{e^{u/2}}{2} + \frac{e^{-u/4}}{2} \\ -\frac{e^{u/2}}{2} - \frac{e^{-u/4}}{2} \end{bmatrix}$$

$$r_2(u) = \frac{1}{4} \cdot e^{\frac{1}{2}u} \cdot \begin{bmatrix} 1 \\ 1 \end{bmatrix} + \frac{1}{4} e^{-\frac{1}{4}u} \begin{bmatrix} -1 \\ 1 \end{bmatrix}$$

$$= \begin{bmatrix} \frac{e^{u/2}}{4} - \frac{e^{-u/4}}{4} \\ \frac{e^{u/2}}{4} + \frac{e^{-u/4}}{4} \end{bmatrix}$$

c) Illustrate with Maple.

```
> r[1]:=u-> <-exp(u/2)/2+exp(-u/4)/2,-exp(u/2)/2-exp(-u/4)/2>:
  r[1](u):
> r[2]:=u-> <exp(u/2)/4-exp(-u/4)/4,exp(u/2)/4+exp(-u/4)/4>:
  r[2](u):
> field:=fieldplot(V,x=-2..2,y=-2..2):
  c1:=plot([vop(r[1](u)),u=-10..10],scaling=constrained,thickness=3):
  c2:=plot([vop(r[2](u)),u=-10..10],scaling=constrained,thickness=3):
> display(field,c1,c2,view=[-2..2,-2..2]);
```



Ex 2. Let $V(x,y,z) = (z, \frac{1}{10}y, -x)$.

a) Compute eigenvalues of $\underline{\underline{A}}$ corresponding to V.

$$\underline{\underline{A}} = \begin{bmatrix} 0 & 0 & 1 \\ 0 & \frac{1}{10} & 0 \\ -1 & 0 & 0 \end{bmatrix}$$

```
> A:=<0,0,1; 0,1/10,0; -1,0,0>;
```

$$A := \begin{bmatrix} 0 & 0 & 1 \\ 0 & \frac{1}{10} & 0 \\ -1 & 0 & 0 \end{bmatrix}$$

```
> Eigenvalues(A);
```

$$\begin{bmatrix} \frac{1}{10} \\ \mathrm{I} \\ -\mathrm{I} \end{bmatrix}$$

b) Determine the flow curve $r(u)$ for which $r(0) = (1,1,1)$.

Let's just use Maple.

```
> l1:=diff(x(u),u)=z(u);
  l2:=diff(y(u),u)=y(u)/10;
  l3:=diff(z(u),u)=-x(u);
```

$$l1 := \frac{d}{du}\, x(u) = z(u)$$

$$l2 := \frac{d}{du}\, y(u) = \frac{y(u)}{10}$$

$$l3 := \frac{d}{du}\, z(u) = -x(u)$$

```
> dsolve({l1,l2,l3,x(0)=1,y(0)=1,z(0)=1},{x(u),y(u),z(u)});
```

$$\left\{x(u) = \sin(u) + \cos(u),\, y(u) = e^{\frac{u}{10}},\, z(u) = \cos(u) - \sin(u)\right\}$$

```
> r:=u-> <sin(u) + cos(u),exp(u/10),cos(u) - sin(u)>:
  r(u);
```

$$\begin{bmatrix} \sin(u) + \cos(u) \\ e^{\frac{u}{10}} \\ \cos(u) - \sin(u) \end{bmatrix}$$

c) Illustrate with Maple.

```
> c1:=spacecurve(r(u),u=0..5*Pi,thickness=5,color=brown):
  field:=fieldplot3d(V(x,y,z),x=-2..2,y=1..5,z=-2..2,grid=[5,5,5]):
  display(c1,field,scaling=constrained,axes=normal);
```



Consider the line segment $\mathcal{L}$ from $(1,1,1)$ to $(2,2,2)$.

d) Provide a parametric representation of $\mathcal{L}$.

$$\begin{bmatrix}1\\1\\1\end{bmatrix} + v\begin{bmatrix}1\\1\\1\end{bmatrix}, \quad v \in [0,1].$$

e) Determine the flow curve $s(u)$ of V, where $s(0)$ is an arbitrary point on $\mathcal{L}$.

```
> dsolve({l1,l2,l3,x(0)=1+v,y(0)=1+v,z(0)=1+v},{x(u),y(u),z(u)});
```

$$\left\{x(u) = (1+v)\sin(u) + (1+v)\cos(u), y(u) = (1+v)\,e^{\frac{u}{10}}, z(u) = (1+v)\cos(u) - (1+v)\sin(u)\right\}$$

```
> s:=u-> <(1 + v)*sin(u) + (1 + v)*cos(u), (1 + v)*exp(u/10), (1 + v)*cos(u) - (1 + v)*sin(u)>:
  s(u);
```

$$\begin{bmatrix}(1+v)\sin(u) + (1+v)\cos(u)\\(1+v)\,e^{\frac{u}{10}}\\(1+v)\cos(u) - (1+v)\sin(u)\end{bmatrix}$$

f) Provide a parametric representation $k(u,v)$ of the surface $\mathcal{F}$ that $\mathcal{L}$ sweeps/flows through for $u \in [0, 5\pi]$. Provide a plot.

```
> s:=(u,v)-> <(1 + v)*sin(u) + (1 + v)*cos(u), (1 + v)*exp(u/10), (1 + v)*cos(u) - (1 + v)*sin(u)>:
  s(u,v);
```

$$\begin{bmatrix}(1+v)\sin(u) + (1+v)\cos(u)\\(1+v)\,e^{\frac{u}{10}}\\(1+v)\cos(u) - (1+v)\sin(u)\end{bmatrix}$$

```
> F:=plot3d(s(u,v),u=0..5*Pi,v=0..1,style=surface):
  field1:=fieldplot3d(V(x,y,z),x=-2..2,y=1..9,z=-2..2,grid=[5,5,5]):
  display(field1,F);
```



Ex 3. Let $f(x,y,z) = (x-1)^2 + 2(y-1)^2 + (z-1)^2 - 4$.

a) Compute ∇f.

$$\nabla f(x,y,z) = \begin{bmatrix} 2x-2 \\ 4y-4 \\ 2z-2 \end{bmatrix}$$

b) Parametrize K_0 corresponding to $f(x,y,z) = 0$.

$$f(x,y,z) = (x-1)^2 + 2(y-1)^2 + (z-1)^2 - 4 = 0$$

$$\Leftrightarrow \quad (x-1)^2 + 2(y-1)^2 + (z-1)^2 = 4$$

$$\Leftrightarrow \quad (x,y,z) = (0,0,0)$$

We see that K_0 is $(x-1)^2 + 2(y-1)^2 + (z-1)^2 = 4$, which according to section 22.3 is an ellipsoid. The center is $(1,1,1)$ and the semi axes are $a = 2$, $b = \sqrt{2}$ and $c = 2$.

Now set

$$r(u,v) = \begin{bmatrix} 2 \sin u \cdot \cos v \\ \sqrt{2} \sin u \cdot \sin v \\ 2 \cos u \end{bmatrix} + \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}, \quad u \in [0,\pi], \; v \in [0, 2\pi].$$

c)

```
> V:=(x,y,z)-><2*x-2,4*y-4,2*z-2>:
  V(x,y,z):
> r:=(u,v)-> <2*sin(u)*cos(v)+1,sqrt(2)*sin(u)*sin(v)+1,2*cos(u)+1>:
  r(u,v):
> surf:=plot3d(r(u,v),u=0..Pi,v=0..2*Pi,scaling=constrained):
  field:=fieldplot3d(V(x,y,z),x=-1..3,y=-1..3,z=-1..3):
  display(surf,field);
```



d) Show that ∇f is perpendicular to K_0 (to the tangent at all points P on K_0).

```
> ru:=diff(r(u,v),u):rv:=diff(r(u,v),v):
> prik(V(vop(r(u,v))),ru):
  simplify(%);
  prik(V(vop(r(u,v))),rv):
  simplify(%);
```

$$0$$
$$0$$

f) "The gradient points in the direction of highest growth". What does it mean in this context?

Essentially we have that the gradient is parallel to the normal of the level set.

Ex4. Let

$$U(x,y,z) = \begin{bmatrix} xy \cdot \cos z \\ y^2 + xz \\ 3z \end{bmatrix} \quad \text{and} \quad V(x,y,z) = \begin{bmatrix} 2xe^{x^2} \\ 2\cos(y^2)\cdot y \\ 3 \end{bmatrix}.$$

a) Which one is a gradient vector field, and which is not?

We need a smooth function f for which ∇f is equal to the vector fields above.

Since this implies $f''_{xy} = f''_{yx}$ we find

$$\frac{d}{dy} xy \cdot \cos z = x \cdot \cos z \neq z = \frac{d}{dx} y^2 + xz,$$

so U is not a gradient vector field. In the case of V we have a smooth function

$$f(x,y,z) = \begin{bmatrix} e^{x^2} \\ \sin(y^2) \\ 3z \end{bmatrix} \quad \text{for which} \quad \nabla f(x,y,z) = V(x,y,z).$$

Ex 5. Let

$$V(x,y) = \begin{bmatrix} x^2 - 2xy \\ y^2 - 2xy \end{bmatrix}$$

and K be given by

$$y = x^2, \quad x \in [-1,1].$$

a) Compute $\int_K V \cdot \underline{e} \, d\mu$.

The curve can be represented by $r(u) = \begin{bmatrix} u \\ u^2 \end{bmatrix}$, $u \in [-1,1]$.

Now we have

$$\int_K V \cdot \underline{e} \, d\mu = \int_{-1}^{1} V(r(u)) \cdot r'(u) \, du$$

$$= \int_{-1}^{1} \begin{bmatrix} -2u^3 + u^2 \\ u^4 - 2u^3 \end{bmatrix} \cdot \begin{bmatrix} 1 \\ 2u \end{bmatrix} du$$

$$= \int_{-1}^{1} 2u^5 - 4u^4 - 2u^3 + u^2 \, du$$

$$= \left[\frac{1}{3}u^6 - \frac{4}{5}u^5 - \frac{1}{2}u^4 + \frac{1}{3}u^3 \right]_{-1}^{1}$$

$$= \frac{1}{3} - \frac{4}{5} - \frac{1}{2} + \frac{1}{3} - \left(\frac{1}{3} + \frac{4}{5} - \frac{1}{2} - \frac{1}{3} \right)$$

$$= -\frac{19}{30} - \frac{9}{30} = -\frac{28}{30} = -\frac{14}{15}$$

Now we're given $U(x,y,z) = \begin{bmatrix} y^2 - z^2 \\ 2yz \\ -z^2 \end{bmatrix}$ and a curve G paramatrized as

$$S(u) = \begin{bmatrix} u \\ u^2 \\ u^3 \end{bmatrix}, \quad u \in [-1,1].$$

b) Compute the tangential line integral $\int_{\mathcal{G}} \mathbf{u} \cdot \mathbf{e} \, d\mu$.

$$\int_{-1}^{1} \mathbf{u}(s(u)) \cdot s'(u) \, du = \int_{-1}^{1} \begin{bmatrix} u^4 - u^6 \\ 2u^5 \\ -u^2 \end{bmatrix} \cdot \begin{bmatrix} 1 \\ 2u \\ 3u^2 \end{bmatrix} du$$

$$= \int_{-1}^{1} 3u^6 - 2u^4 \, du$$

$$= \left[\frac{3}{7} u^7 - \frac{2}{5} u^5 \right]_{-1}^{1}$$

$$= \frac{1}{35} - \left(-\frac{1}{35} \right) = \frac{2}{35}$$

Ex 6. Let $V(x,y) = \begin{bmatrix} xy \\ x \end{bmatrix}$.

a) Compute the tangential line integral of V along the straight line to $P = (x,y)$.

$$r(u) = u \begin{bmatrix} x \\ y \end{bmatrix}, \quad u \in [0,1].$$

We have $V(r(u)) = \begin{bmatrix} u^2 x y \\ u x \end{bmatrix}$ and $r'(u) = \begin{bmatrix} x \\ y \end{bmatrix}$.

```
> V:=(x,y)-> <x*y,x>:
  r:=u->u*<x,y>:
  ru:=diff(r(u),u):
  int(prik(V(vop(r(u))),ru),u=0..1);
```

$$\frac{1}{3} x^2 y + \frac{1}{2} y x$$

b) Sketch a stair line.



c) Compute the tangential line integral of V along a stair line.

$$r_1(u) = \begin{bmatrix} u \\ 0 \end{bmatrix}, \quad \text{where} \quad u \in [0, x].$$

$$r_2(u) = \begin{bmatrix} x \\ u \end{bmatrix}, \quad \text{where} \quad u \in [0, y].$$

$$\int_T V \cdot \underline{e} \, d\mu = \int_0^x V(r_1(u)) \cdot r_1'(u) \, du + \int_0^y V(r_2(u)) \cdot r_2'(u) \, du$$

```
> r1:=u-> <u,0>:
  r2:=u-> <x,u>:
  r1u:=diff(r1(u),u):
  r2u:=diff(r2(u),u):
> int(prik(V(vop(r1(u))),r1u),u=0..x)+
  int(prik(V(vop(r2(u))),r2u),u=0..y);
```

$$y x$$

d) Is V a gradient vector field?

No a gradient vector field is independent of the path traversed to the point P, and

$$\frac{1}{3} x^2 y + \frac{1}{2} y x \neq y x.$$

This is also stated in proposition 27.10.

Ex 7. V is a gradient vector field if and only if it has a scalar potential.

Let

$$V(x, y, z) = \begin{bmatrix} y \cos(xy) \\ z + x \cos(xy) \\ y \end{bmatrix}.$$

a) Determine the tangential line integral of V to $P = (x, y, z)$.

```
> r1:=u-> <u,0,0>:
  r2:=u-> <x,u,0>:
  r3:=u-> <x,y,u>:
  r1u:=diff(r1(u),u):
  r2u:=diff(r2(u),u):
  r3u:=diff(r3(u),u):
> int(prik(V(vop(r1(u))),r1u),u=0..x)+
  int(prik(V(vop(r2(u))),r2u),u=0..y)+
  int(prik(V(vop(r3(u))),r3u),u=0..z);
```

$$\sin(yx) + yz$$

b) Investigate whether V is a gradient vector field.

```
> F:=unapply(%,[x,y,z]):
  F(x,y,z);
```

$$\sin(yx) + yz$$

```
> grad(F(x,y,z),[x,y,z]);
```

$$\begin{bmatrix} y\cos(yx) \\ z + x\cos(yx) \\ y \end{bmatrix}$$

This indeed works! For a general solution we may add a constant.

Let $U(x,y,z) = \dfrac{1}{1+x^2y^2+2xyz^2+z^4}\begin{bmatrix} y \\ x \\ 2z \end{bmatrix}$

c) Investigate whether U is a gradient vector field.

```
> g:=(x,y,z)-> 1+x^2*y^2+2*x*y*z^2+z^4:
  g(x,y,z):
> U:=(x,y,z)-> 1/g(x,y,z)*<y,x,2*z>:
  U(x,y,z);
```

$$\begin{bmatrix} \frac{y}{x^2y^2+2xyz^2+z^4+1} \\ \frac{x}{x^2y^2+2xyz^2+z^4+1} \\ \frac{2z}{x^2y^2+2xyz^2+z^4+1} \end{bmatrix}$$

```
> r:=u-> u*<x,y,z>:
  ru:=diff(r(u),u):
> int(prik(U(vop(r(u))),ru),u=0..1):
  simplify(%);
```

$$\arctan(yx+z^2)$$

```
> F:=unapply(%,(x,y,z)):
  F(x,y,z);
  grad(F(x,y,z),[x,y,z]);
```

$$\arctan(yx+z^2)$$

$$\begin{bmatrix} \frac{y}{(yx+z^2)^2+1} \\ \frac{x}{(yx+z^2)^2+1} \\ \frac{2z}{(yx+z^2)^2+1} \end{bmatrix}$$

```
> grad(F(x,y,z),[x,y,z])-U(x,y,z):
  simplify(%);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Yes indeed we found a scalar potential!